

NAG Fortran Library Routine Document

E01ABF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

E01ABF interpolates at a given point x from a table of function values evaluated at equidistant points, by Everett's formula.

2 Specification

```
SUBROUTINE E01ABF(N, P, A, G, N1, N2, IFAIL)
INTEGER          N, N1, N2, IFAIL
real           P, A(N1), G(N2)
```

3 Description

This routine interpolates at a given point

$$x = x_0 + ph, \quad \text{where} \quad -1 < p < 1$$

from a table of values $(x_0 + mh)$ and y_m where $m = -(n-1), -(n-2), \dots, -1, 0, 1, \dots, n$. The formula used is that of Fröberg (1970), neglecting the remainder term:

$$y_p = \sum_{r=0}^{n-1} \left(\frac{1-p+r}{2r+1} \right) \delta^{2r} y_0 + \sum_{r=0}^{n-1} \left(\frac{p+r}{2r+1} \right) \delta^{2r} y_1.$$

The values of $\delta^{2r} y_0$ and $\delta^{2r} y_1$ are stored on exit from the routine in addition to the interpolated function value y_p .

4 References

Fröberg C E (1970) *Introduction to Numerical Analysis* Addison-Wesley

5 Parameters

- 1: N – INTEGER *Input*
On entry: n , half the number of points to be used in the interpolation.
- 2: P – *real* *Input*
On entry: the point p at which the interpolated function value is required i.e., $p = (x - x_0)/h$ with $-1.0 < p < 1.0$.
Constraint: $-1.0 < P < 1.0$.
- 3: A(N1) – *real* array *Input/Output*
On entry: A(i) must be set to the function value y_{i-n} for $i = 1, 2, \dots, 2n$.
On exit: the contents of A are unspecified.

- 4: G(N2) – *real* array *Output*
On exit: the array contains
 y_0 in G(1)
 y_1 in G(2)
 $\delta^{2r}y_0$ in G(2r + 1)
 $\delta^{2r}y_1$ in G(2r + 2) for $r = 1, 2, \dots, n - 1$.
 The interpolated function value y_p is stored in G(2n + 1).
- 5: N1 – INTEGER *Input*
On entry: the value $2n$, that is, N1 is equal to the number of data points.
- 6: N2 – INTEGER *Input*
On entry: the value $2n + 1$, that is, N2 is one more than the number of data points.
- 7: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.
On exit: IFAIL = 0 unless the routine detects an error (see Section 6).
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, $P \leq -1.0$,
 or $P \geq 1.0$.

7 Accuracy

In general, increasing n improves the accuracy of the result until full attainable accuracy is reached, after which it might deteriorate. If x lies in the central interval of the data (i.e., $0.0 \leq p \leq 1.0$), as is desirable, an upper bound on the contribution of the highest order differences (which is usually an upper bound on the error of the result) is given approximately in terms of the elements of the array G by $a \times (|G(2n - 1)| + |G(2n)|)$, where $a = 0.1, 0.02, 0.005, 0.001, 0.0002$ for $n = 1, 2, 3, 4, 5$ respectively, thereafter decreasing roughly by a factor of 4 each time.

8 Further Comments

The computation time increases as the order of n increases.

9 Example

To interpolate at the point $x = 0.28$ from the function values

$$\begin{pmatrix} x_i & -1.00 & -0.50 & 0.00 & 0.50 & 1.00 & 1.50 \\ y_i & 0.00 & -0.53 & -1.00 & -0.46 & 2.00 & 11.09 \end{pmatrix}.$$

We take $n = 3$ and $p = 0.56$.

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      E01ABF Example Program Text
*      Mark 14 Revised.  NAG Copyright 1989.
*      .. Parameters ..
      INTEGER          NMAX, N1MAX, N2MAX
      PARAMETER       (NMAX=10,N1MAX=2*NMAX,N2MAX=2*NMAX+1)
      INTEGER          NIN, NOUT
      PARAMETER       (NIN=5,NOUT=6)
*      .. Local Scalars ..
      real            P
      INTEGER          I, IFAIL, N, R
*      .. Local Arrays ..
      real            A(N1MAX), G(N2MAX)
*      .. External Subroutines ..
      EXTERNAL        E01ABF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'E01ABF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, P
      IF (N.GT.0 .AND. N.LE.NMAX) THEN
        READ (NIN,*) (A(I),I=1,2*N)
        IFAIL = 0
*
        CALL E01ABF(N,P,A,G,2*N,2*N+1,IFAIL)
*
        WRITE (NOUT,*)
        DO 20 R = 0, N - 1
          WRITE (NOUT,99999) 'Central differences order ', R,
+           ' of Y0 =', G(2*R+1)
          WRITE (NOUT,99998) '                               Y1 =',
+           G(2*R+2)
20      CONTINUE
        WRITE (NOUT,*)
        WRITE (NOUT,99998) 'Function value at interpolation point =',
+           G(2*N+1)
        END IF
      STOP
*
99999 FORMAT (1X,A,I1,A,F12.5)
99998 FORMAT (1X,A,F12.5)
      END
```

9.2 Program Data

```
E01ABF Example Program Data
 3      0.56
 0.00  -0.53  -1.00  -0.46  2.00  11.09
```

9.3 Program Results

E01ABF Example Program Results

```
Central differences order 0 of Y0 = -1.00000
                                Y1 = -0.46000
Central differences order 1 of Y0 =  1.01000
                                Y1 =  1.92000
Central differences order 2 of Y0 = -0.04000
                                Y1 =  3.80000

Function value at interpolation point = -0.83591
```
